' {$STAMP BS2}

' {$PBASIC 2.5}

'

'

' Robot_Sense_Table_Edge.bs2

'

' ------- I/O Definitions ------------

Ping              PIN       15

' ------- Constants ------------------

Trigger           CON       5          ' trigger pulse = 10uSec

Scale             CON       $200       ' raw * 2.00 = uSec

RawToCm           CON       2257       ' 1 / 29.034 (w/ **)

IsHigh            CON       1          ' for PULSOUT

IsLow             CON       0

' ------- Variables ------------------

rawDist           VAR       Word           ' distance as function of lapsed time

cm                VAR       Word

edgeCtr           VAR       Word

distanceCtr       VAR       Word

edgeCtr     =     0

distanceCtr =     0

' ------- Initialization -------------

Reset:

DEBUG CLS,                             ' setup report screen

  "Parallax Ping))) Sonar ", CR,

  "---------------------- ", CR, CR,

  "Time (uS)......     ", CR,

  "Centimeters....     "

' ------- Program Code --------------

Main:

DO

  GOSUB Get_Sonar

  DEBUG CRSRXY, 15, 3,

        DEC rawDist, CLREOL,

        CRSRXY, 15, 4,

        DEC cm, CLREOL

  PAUSE 100

  GOSUB Move_Robot

LOOP

END

' ------- Subroutines ---------------

Get_Sonar:

  Ping = IsLow                            ' make trigger 0-1-0

  PULSOUT Ping, Trigger                   ' activate sensor

  PULSIN Ping, IsHigh, rawDist            ' measure echo pulse

  PAUSE 20

  rawDist = rawDist */ Scale              ' convert to uSec

  rawDist = rawDist / 2                   ' remove return trip

  cm = rawDist ** RawToCm

  RETURN

Move_Robot:

  IF (cm <= 3 ) THEN

    GOSUB Forward

    edgeCtr = edgeCtr + 1

  ENDIF

  IF ( cm > 3 ) THEN

    GOSUB BackUp

    GOSUB NinetyLeft

    GOSUB NinetyLeft

    GOSUB Get_Sonar

    IF ( cm <= 3 ) THEN

      GOSUB Forward

      edgeCtr = edgeCtr + 1

      IF ( edgeCtr > 2 ) THEN

        GOSUB EndProgram

      ENDIF

    ELSE

      GOSUB EndProgram

    ENDIF

  ENDIF

  RETURN

Forward:

  PULSOUT 12, 662

  PULSOUT 13, 850

  PAUSE 20

  RETURN

BackUp:

  FOR distanceCtr = 0 TO 30

    PULSOUT 13, 700

    PULSOUT 12, 800

    PAUSE 20

  NEXT

  RETURN

NinetyLeft:

  FOR distanceCtr = 0 TO 32

    PULSOUT 12, 662

    PULSOUT 13, 745

    PAUSE 20

  NEXT

  RETURN

EndProgram:

END

